ACAN2515 library for Arduino Version 1.0.2

Pierre Molinaro

November 3, 2018

Contents

| 1 | Versions Features | | |
|-------------|-------------------|--|----|
| 2 | | | |
| 3 Data flow | | | |
| 4 | A si | mple example: LoopBackDemo | 5 |
| 5 | The | CANMessage class | 8 |
| 6 | Con | necting a MCP2515 to your microcontroller | 9 |
| | 6.1 | Using alternate pins on Teensy 3.x | 9 |
| | 6.2 | Using alternate pins on an Adafruit Feather M0 | 11 |
| 7 | Sending frames | | 12 |
| | 7.1 | The tryToSend method | 12 |
| | 7.2 | Driver transmit buffer sizes | 13 |
| | 7.3 | The transmitBufferSize method | 13 |
| | 7.4 | The transmitBufferCount method | 14 |
| | 7.5 | The transmitBufferPeakCount method | 14 |
| 8 | Reti | rieving received messages using the receive method | 14 |
| | 8.1 | Driver receive buffer size | 15 |
| | 8.2 | The receiveBufferSize method | 15 |
| | 8.3 | The receiveBufferCount method | 15 |
| | 8.4 | The receiveBufferPeakCount method | 16 |
| 9 | Acce | eptance filters | 16 |
| | 9.1 | Default behaviour | 16 |
| | 9.2 | Defining filters | 17 |

| CONTENTS | CONTENTS |
|----------|----------|
| | |
| | |

| | 9.2.1 9.2.2 | The state of the s | 19 20 | | | |
|-----------|--------------------------------------|--|----------|--|--|--|
| 10 | The dispat | chReceivedMessage method | 21 | | | |
| 11 | The ACAN2515::begin method reference | | | | | |
| | 11.1 The A | CAN2515::begin method prototypes | 22 | | | |
| | | | 22 | | | |
| | 11.3 The e | rror code | 22 | | | |
| | 11.3.1 | kNoMCP2515 | 23 | | | |
| | 11.3.2 | kTooFarFromDesiredBitRate | 23 | | | |
| | 11.3.3 | kInconsistentBitRateSettings | 23 | | | |
| | 11.3.4 | | 23 | | | |
| | 11.3.5 | | 24 | | | |
| | 11.3.6 | kRequestedModeTimeOut | 24 | | | |
| | 11.3.7 | | 24 | | | |
| | 11.3.8 | kOneFilterMaskRequiresOneOrTwoAcceptanceFilters | 24 | | | |
| | 11.3.9 | | 24 | | | |
| | | | | | | |
| 12 | ACAN2515Settings class reference | | | | | |
| | | 0 | 24 | | | |
| | | Section 3, 1 and 1 | 28 | | | |
| | | | 29 | | | |
| | | | 29 | | | |
| | 12.5 The p | pmFromDesiredBitRate method | 30 | | | |
| | 12.6 The s | • | 31 | | | |
| | 12.7 Prope | erties of the ACAN2515Settings class | 31 | | | |
| | 12.7.1 | The mOneShotModeEnabled property | 31 | | | |
| | 12.7.2 | The mTXBPriority property | 31 | | | |
| | 12.7.3 | The mRequestedMode property | 32 | | | |
| | 12.7.4 | The mCLKOUT property | 32 | | | |
| | 12.7.5 | The mRolloverEnable property | 32 | | | |
| 13 | CAN contr | voller state | 32 | | | |
| | 13.1 The r | eceiveErrorCounter method | 33 | | | |
| | 13.2 The t | ransmitErrorCounter method | 33 | | | |

1 Versions

| Version | Date | Comment | |
|---------|------------------|--|--|
| 1.0.0 | October 12, 2018 | Initial release | |
| 1.0.1 | October 23, 2018 | Workaround external interrupt masking for Teensy 3.5 $/$ 3.6 | |
| | | Use of a lambda function for interrupt service routine | |
| 1.0.2 | November 3, 2018 | Correct setting of rtr and ext properties on message receive | |
| | | (thanks to Arjan-Woltjer for having fixed this bug, https:// | |
| | | github.com/pierremolinaro/acan2515/pull/1) | |

2 Features

The ACAN2515 library is a MCP2515 CAN ("Controller Area Network") Controller driver for any board running Arduino. It has been designed to make it easy to start and to be easily configurable:

- default configuration sends and receives any frame no default filter to provide;
- efficient built-in CAN bit settings computation from user bit rate;
- user can fully define its own CAN bit setting values;
- all reception filter registers are easily defined (2 mask registers, 6 acceptance registers);
- reception filters accept call back functions;
- driver transmit buffer sizes are customisable;
- driver receive buffer size is customisable;
- overflow of the driver receive buffer is detectable;
- loop back, self reception, listing only MCP2515 controller modes are selectable.

3 Data flow

The figure 1 illustrates message flow for sending and receiving CAN messages.

Sending messages. A message is defined by an instance of CANMessage class. For sending a message, user code calls the tryToSend method – see section 7 page 12, and the idx property of the sent message specifies a transmit buffer. The ACAN2515 driver defines 3 transmit buffers, each of them corresponding to the one of the 3 MCP2515 transmit buffers (TXB0, TXB1, TXB2). These buffers can contain at most one message. The message is transferred in a driver transmit before to be moved by the interrupt service routine into the corresponding MCP2515 transmit buffer. The size of the *Driver Transmit Buffer 0* is 16 by default, the size of the *Driver Transmit Buffer 1* and *Driver Transmit Buffer 1* are zero by default – see section 7.2 page 13 for changing the default values.

Receiving messages. The MCP2515 CAN Protocol Engine transmits all correct frames to the reception filters. By default, they are configured as pass-all, see section 9 page 16 for configuring them. Messages




Figure 1 – Message flow in ACAN2515 driver and MCP2515 CAN Controller

that pass the filters are stored in the *Reception Registers* (RXBO and RXB1). The interrupt service routine transfers the messages from these registers to the *Driver Receive Buffer*. The size of the *Driver Receive Buffer* is 32 by default – see section 8.1 page 15 for changing the default value. Three user methods are available:

- the available method returns false if the Driver Receive Buffer is empty, and true otherwise;
- the receive method retrieves messages from the Driver Receive Buffer see section 8 page 14;
- the dispatchReceivedMessage method if you have defined the reception filters that name a callback function see section 10 page 21.

Sequentiality. The ACAN2515 driver and the configuration of the MCP2515 controller can ensure sequentiality of data messages¹, under some conditions. The driver ensures the sequentiality of the emissions, provided that you use only one transmit buffer: if an user program calls tryToSend first for a message M_1 specifying the B_i buffer and then for a message M_2 specifying the same buffer, the driver ensures that M_1 will be sent on the CAN bus before M_2 . However, if M_2 specifies an other buffer, there is

¹Sequentiality means that if an user program calls tryToSend first for a message M_1 and then for a message M_2 , the message M_1 will be allways retrieved by receive or dispatchReceivedMessage before the message M_2 .

no guarantee that M_1 will appear on the bus before M_2 . In reception, the driver ensures sequentiality based on the reception filters: if a received message M_1 passes a given filter, and then a received message M_2 passes the same filter, then the messages are retrieved in this order by the receive or the dispatchReceivedMessage methods.

4 A simple example: LoopBackDemo

The following code is a sample code for introducing the ACAN2515 library, extracted from the LoopBackDemo sample code included in the library distribution. It runs natively on any Arduino compatible board, and is easily adaptable to any microcontroller supporting SPI. It demonstrates how to configure the driver, to send a CAN message, and to receive a CAN message.

Note: this code runs without any CAN transceiver (the TXCAN and RXCAN pins of the MCP2515 are left open), the MCP2515 is configured with the *loop back* setting on.

```
#include <ACAN2515.h>
```

This line includes the ACAN2515 library.

```
static const byte MCP2515_SCK = 27 ; // SCK input of MCP2515
static const byte MCP2515_SI = 28 ; // SI input of MCP2515
static const byte MCP2515_SO = 39 ; // SO output of MCP2515
```

Define the SPI alternate pins. This is actually required if you uses SPI alternate pins.

```
static const byte MCP2515_CS = 20 ; // CS input of MCP2515
static const byte MCP2515_INT = 37 ; // INT output of MCP2515
```

Define the pins connected to \overline{CS} and \overline{INT} pins.

```
ACAN2515 can (MCP2515_CS, SPI, MCP2515_INT);
```

Instanciation of the ACAN2515 library, declaration and initialization of the can object that implements the driver. The constructor names: the number of the pin connected to the $\overline{\tt CS}$ pin, the SPI object (you can use SPI1, SPI2, ...), the number of the pin connected to the $\overline{\tt INT}$ pin.

```
static const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
```

Specifies the frequency of the MCP2515 quartz.

```
void setup () {
//--- Switch on builtin led
```

```
pinMode (LED_BUILTIN, OUTPUT) ;
  digitalWrite (LED_BUILTIN, HIGH) ;

//--- Start serial
  Serial.begin (38400) ;

//--- Wait for serial (blink led at 10 Hz during waiting)
  while (!Serial) {
    delay (50) ;
    digitalWrite (LED_BUILTIN, !digitalRead (LED_BUILTIN)) ;
}
```

Builtin led is used for signaling. It blinks led at 10 Hz during until serial monitor is ready.

```
SPI.begin ();
```

You should call SPI.begin. Many platforms define alternate pins for SPI. On Teensy 3.x (section 6.1 page 9), selecting alternate pins should be done before calling SPI.begin, on Adafruit Feather M0 (section 6.2 page 11), this should be done after. Calling SPI.begin explicitly allows you to fully handle alternate pins.

```
ACAN2515Settings settings (QUARTZ_FREQUENCY, 125 * 1000);
```

Configuration is a four-step operation. This line is the first step. It instanciates the settings object of the ACAN2515Settings class. The constructor has two parameters: the MCP2515 quartz frequency, and the desired CAN bit rate (here, 125 kb/s). It returns a settings object fully initialized with CAN bit settings for the desired bit rate, and default values for other configuration properties.

```
settings.mRequestedMode = ACAN2515RequestedMode::LoopBackMode ;
```

This is the second step. You can override the values of the properties of settings object. Here, the mRequestedMode property is set to LoopBackMode – its value is NormalMode by default. Setting this property enables *loop back*, that is you can run this demo sketch even it you have no connection to a physical CAN network. The section 12.7 page 31 lists all properties you can override.

```
const uint32_t errorCode = can.begin (settings, [] { can.isr () ; }) ;
```

This is the third step, configuration of the can driver with settings values. The driver is configured for being able to send any (standard / extended, data / remote) frame, and to receive all (standard / extended, data / remote) frames. If you want to define reception filters, see section 9 page 16. The second argument is the *interrupt service routine*, and is defined by a C++ lambda expression². See section 11.2 page 22 for using a function instead.

²https://en.cppreference.com/w/cpp/language/lambda

```
if (errorCode != 0) {
    Serial.print ("Configurationuerroru0x") ;
    Serial.println (errorCode, HEX) ;
}
```

Last step: the configuration of the can driver returns an error code, stored in the errorCode constant. It has the value 0 if all is ok – see section 11.3 page 22.

```
static unsigned gBlinkLedDate = 0 ;
static unsigned gReceivedFrameCount = 0 ;
static unsigned gSentFrameCount = 0 ;
```

The gSendDate global variable is used for sending a CAN message every 2 s. The gSentCount global variable counts the number of sent messages. The gReceivedCount global variable counts the number of received messages.

```
void loop() {
   CANMessage frame ;
```

The message object is fully initialized by the default constructor, it represents a standard data frame, with an identifier equal to 0, and without any data – see section 5 page 8.

```
if (gBlinkLedDate < millis ()) {
    gBlinkLedDate += 2000 ;
    digitalWrite (LED_BUILTIN, !digitalRead (LED_BUILTIN)) ;
    const bool ok = can.tryToSend (frame) ;
    if (ok) {
        gSentFrameCount += 1 ;
        Serial.print ("Sent:_u") ;
        Serial.println (gSentFrameCount) ;
    }else{
        Serial.println ("Send_failure") ;
    }
}</pre>
```

We try to send the data message. Actually, we try to transfer it into the *Driver transmit buffer*. The transfer succeeds if the buffer is not full. The tryToSend method returns false if the buffer is full, and true otherwise. Note the returned value only tells if the transfer into the *Driver transmit buffer* is successful or not: we have no way to know if the frame is actually sent on the the CAN network. Then, we act the successfull transfer by setting gSendDate to the next send date and incrementing the gSentCount variable. Note if the transfer did fail, the send date is not changed, so the tryToSend method will be called on the execution of the loop function.

```
if (can.available ()) {
   can.receive (frame);
   gReceivedFrameCount ++;
   Serial.print ("Received:");
   Serial.println (gReceivedFrameCount);
}
```

As the MCP2515 controller is configured in *loop back* mode, all sent messages are received. The receive method returns false if no message is available from the *driver reception buffer*. It returns true if a message has been successfully removed from the *driver reception buffer*. This message is assigned to the message object. If a message has been received, the gReceivedCount is incremented and displayed.

5 The CANMessage class

Note. The CANMessage class is declared in the CANMessage.h header file. The class declaration is protected by an include guard that causes the macro GENERIC_CAN_MESSAGE_DEFINED to be defined. The ACAN³ (version 1.0.3 and above) driver, the ACAN²2517⁴ driver contain an identical CANMessage.h file header, enabling using ACAN driver, ACAN²2515 driver and ACAN²2517 driver in a same sketch.

A *CAN message* is an object that contains all CAN frame user informations. All properties are initialized by default, and represent a standard data frame, with an identifier equal to 0, and without any data.

Note the message datas are defined by an union. So message datas can be seen as height bytes, four 16-bit unsigned integers, two 32-bit, or one 64-bit. Be aware that multi-byte integers are subject to endianness (Cortex M4 processors of Teensy 3.x are little-endian).

The idx property is not used in CAN frames, but:

³The ACAN driver is a CAN driver for FlexCAN modules integrated in the Teensy 3.x microcontrollers, https://github.com/pierremolinaro/acan.

⁴The ACAN2517 driver is a CAN driver for the MCP2517 CAN controller, https://github.com/pierremolinaro/acan2517.

- for a received message, it contains the acceptance filter index (see section 10 page 21);
- on sending messages, it is used for selecting the transmit buffer (see section 7.1 page 12).

6 Connecting a MCP2515 to your microcontroller

Connecting a MCP2515 requires 5 pins (figure 2):

- hardware SPI requires you use dedicaced pins of your microcontroller. You can use alternate pins (see below), and if your microcontroller supports several hardware SPIs, you can select any of them;
- connecting the $\overline{\tt CS}$ signal requires one digital pin, that the driver configures as an OUTPUT;
- connecting the $\overline{\tt INT}$ signal requires one other digital pin, that the driver configures as an external interrupt input; so this pin should have interrupt capability (checked by the begin method of the driver object).




Figure 2 - MCP2515 connection to a microcontroller

The begin function of ACAN2515 library configures the selected SPI with a frequency of 10 Mbit/s (the maximum frequency supported by the MCP2515). More precisely, the SPI library of your microcontroller may adopt a frequency lower than 10 Mbit/s; for example, the maximum frequency of the Arduino Uno SPI is 8 Mbit/s.

6.1 Using alternate pins on Teensy 3.x

Demo sketch: LoopBackDemoTeensy3x.

On Teensy 3.x, "the main SPI pins are enabled by default. SPI pins can be moved to their alternate position with SPI.setMOSI(pin), SPI.setMISO(pin), and SPI.setSCK(pin). You can move all of them, or just the ones that conflict, as you prefer."⁵

For example, the LoopBackDemoTeensy3x sketch uses SPIO on a Teensy 3.5 with these alternate pins⁶:

You call the SPI.setMOSI, SPI.setMISO, and SPI.setSCK functions before calling the begin function of your ACAN2515 instance (generally done in the setup function):

```
ACAN2515 can (MCP2515_CS, SPI, MCP2515_INT);
...
```

 $^{^5\}mathrm{See}$ https://www.pjrc.com/teensy/td_libs_SPI.html

⁶See https://www.pjrc.com/teensy/pinout.html




Figure 3 – Using SPI alternate pins on a Teensy 3.5

```
static const byte MCP2515_SCK = 27 ; // SCK input of MCP2515
static const byte MCP2515_SI = 28 ; // SI input of MCP2515
static const byte MCP2515_SO = 39 ; // SO output of MCP2515
...
void setup () {
    ...
    SPI.setMOSI (MCP2515_SI) ;
    SPI.setSCK (MCP2515_SO) ;
    SPI.begin () ;
    ...
    const uint32_t errorCode = can.begin (settings, [] { can.isr () ; }) ;
    ...
```

Note you can use the SPI.pinIsMOSI, SPI.pinIsMISO, and SPI.pinIsSCK functions to check if the alternate pins you select are valid:

```
void setup () {
 Serial.print ("Using pin, #");
 Serial.print (MCP2515_SI) ;
 Serial.print ("uforuMOSI:u");
 Serial.println (SPI.pinIsMOSI (MCP2515_SI) ? "yes" : "NO!!!") ;
 Serial.print ("Using_pin_#");
 Serial.print (MCP2515_S0) ;
 Serial.print ("uforuMISO:u");
 Serial.println (SPI.pinIsMISO (MCP2515_SO) ? "yes" : "NO!!!") ;
 Serial.print ("Using | pin | #");
 Serial.print (MCP2515_SCK) ;
 Serial.print ("uforuSCK:u");
 Serial.println (SPI.pinIsSCK (MCP2515_SCK) ? "yes" : "NO!!!") ;
 SPI.setMOSI (MCP2515_SI) ;
 SPI.setMISO (MCP2515_SO) ;
 SPI.setSCK (MCP2515_SCK) ;
 SPI.begin ();
 const uint32_t errorCode = can.begin (settings, [] { can.isr () ; }) ;
```

6.2 Using alternate pins on an Adafruit Feather M0

 ${\bf Demo~sketch:}~{\tt LoopBackDemoAdafruitFeatherMO}.$

See https://learn.adafruit.com/using-atsamd21-sercom-to-add-more-spi-i2c-serial-ports/overview document that explains in details how configure and set an alternate SPI on Adafruit Feather M0

For example, the LoopBackDemoAdafruitFeatherM0 sketch uses SERCOM1 on an Adafruit Feather M0 as illustrated in figure 4.




Figure 4 – Using SPI alternate pins on an Adafruit Feather M0

The configuration code is the following. Note you should call the pinPeripheral function after calling the mySPI.begin function.

```
#include <wiring_private.h>
. . .
static const byte MCP2515_SCK = 12; // SCK pin, SCK input of MCP2515
static const byte MCP2515_SI = 11; // MOSI pin, SI input of MCP2515
static const byte MCP2515_SO = 10; // MISO pin, SO output of MCP2515
SPIClass mySPI (&sercom1,
                MCP2515_SO, MCP2515_SI, MCP2515_SCK,
                SPI_PAD_0_SCK_3, SERCOM_RX_PAD_2);
static const byte MCP2515_CS = 6 ; // CS input of MCP2515
static const byte MCP2515_INT = 5; // INT output of MCP2515
ACAN2515 can (MCP2515_CS, mySPI, MCP2515_INT);
void setup () {
 mySPI.begin ();
 pinPeripheral (MCP2515_SI, PIO_SERCOM);
 pinPeripheral (MCP2515_SCK, PIO_SERCOM);
 pinPeripheral (MCP2515_SO, PIO_SERCOM);
  const uint32_t errorCode = can.begin (settings, [] { can.isr () ; }) ;
```

7 Sending frames

The ACAN2515 driver define three transmit buffers, each of them corresponding to a MCP2515 hardware buffer.

7.1 The tryToSend method

```
CANMessage message;

// Setup message

const bool ok = can.tryToSend (message);

...
```

You call the tryToSend method for sending a message in the CAN network. Note this function returns before the message is actually sent; this function only appends the message to a transmit buffer.

The idx field of the message specifies the transmit buffer $(0 \to \text{transmit buffer } 0, 1 \to \text{transmit buffer } 1, 2 \to \text{transmit buffer } 2$, any other value $\to \text{transmit buffer } 0$). The default value of the idx field is zero: the message is sent throught TXBO.

The method tryToSend returns:

- true if the message has been successfully transmitted to driver transmit buffer; note that does not mean that the CAN frame has been actually sent;
- false if the message has not been successfully transmitted to driver transmit buffer, it was full.

So it is wise to systematically test the returned value.

A way is to use a global variable to note if the message has been successfully transmitted to driver transmit buffer. For example, for sending a message every 2 seconds:

```
static unsigned gSendDate = 0 ;

void loop () {
   if (gSendDate < millis ()) {
      CANMessage message ;
      // Initialize message properties
      const bool ok = can.tryToSend (message) ;
      if (ok) {
            gSendDate += 2000 ;
      }
   }
}</pre>
```

An other hint to use a global boolean variable as a flag that remains true while the message has not been sent.

```
static bool gSendMessage = false ;

void loop () {
    ...
    if (frame_should_be_sent) {
        gSendMessage = true ;
    }
    ...
    if (gSendMessage) {
        CANMessage message ;
        // Initialize message properties
        const bool ok = can.tryToSend (message) ;
        if (ok) {
            gSendMessage = false ;
        }
    }
    ...
}
```

7.2 Driver transmit buffer sizes

By default:

- driver transmit buffer 0 size is 16;
- driver transmit buffer 1 and 2 sizes are 0.

You can change the default values by setting the mTransmitBufferOSize, mTransmitBuffer1Size, mTransmitBuffer2Size properties of settings variable; for example:

```
ACAN2515Settings settings (QUARTZ_FREQUENCY, 125 * 1000);
settings.mTransmitBufferOSize = 30;
const uint32_t errorCode = can.begin (settings, [] { can.isr (); });
...
```

A zero size is valid: calling the tryToSend method returns true if the corresponding TXBi register is empty, and false if it is full.

7.3 The transmitBufferSize method

The transmitBufferSize method has one argument, the index i of a driver transmit buffer $(0 \le i \le 2)$. It returns the allocated size of this driver transmit buffer, that is the value of settings.mTransmitBufferiSize when the begin method is called.

```
const uint32_t s = can.transmitBufferSize (1); // Driver transmit buffer 1
```

7.4 The transmitBufferCount method

The transmitBufferCount method has one argument, the index i of a driver transmit buffer $(0 \le i \le 2)$. It returns the current number of messages in the driver transmit buffer i.

```
const uint32_t n = can.transmitBufferCount (0); // Driver transmit buffer 0
```

7.5 The transmitBufferPeakCount method

The transmitBufferPeakCount method has one argument, the index i of a driver transmit buffer $(0 \le i \le 2)$. It returns the peak value of message count in the driver transmit buffer i.

```
const uint32_t max = can.transmitBufferPeakCount (2); // Driver transmit buffer 0
```

If the transmit buffer is full when tryToSend is called, the return value of this call is false. In such case, the following calls of transmitBufferPeakCount(i) will return transmitBufferSize (i)+1.

So, when transmitBufferPeakCount(i) returns a value lower or equal to transmitBufferSize (i), it means that calls to tryToSend have allways returned true, and no overflow occurs on transmit buffer i.

8 Retrieving received messages using the receive method

There are two ways for retrieving received messages:

- using the receive method, as explained in this section;
- using the dispatchReceivedMessage method (see section 10 page 21).

This is a basic example:

```
void loop () {
   CANMessage message ;
   if (can.receive (message)) {
      // Handle received message
   }
   ...
}
```

The receive method:

- returns false if the driver receive buffer is empty, message argument is not modified;
- returns true if a message has been has been removed from the driver receive buffer, and the message argument is assigned.

You need to manually dispatch the received messages. If you did not provide any receive filter, you should check the rtr bit (remote or data frame?), the ext bit (standard or extended frame), and the id (identifier value). The following snippet dispatches three messages:

```
void loop () {
   CANMessage message ;
   if (can.receive (message)) {
      if (!message.rtr && message.ext && (message.id == 0x123456)) {
        handle_myMessage_0 (message) ; // Extended data frame, id is 0x123456
      }else if (!message.rtr && !message.ext && (message.id == 0x234)) {
        handle_myMessage_1 (message) ; // Standard data frame, id is 0x234
      }else if (message.rtr && !message.ext && (message.id == 0x542)) {
        handle_myMessage_2 (message) ; // Standard remote frame, id is 0x542
      }
    }
   ...
}
```

The handle_myMessage_0 function has the following header:

```
void handle_myMessage_0 (const CANMessage & inMessage) {
    ...
}
```

So are the header of the handle_myMessage_1 and the handle_myMessage_2 functions.

8.1 Driver receive buffer size

By default, the driver receive buffer size is 32. You can change it by setting the mReceiveBufferSize property of settings variable before calling the begin method:

```
ACAN2515Settings settings (QUARTZ_FREQUENCY, 125 * 1000);
settings.mReceiveBufferSize = 100;
const uint32_t errorCode = can.begin (settings, [] { can.isr (); });
...
```

As the size of CANMessage class is 16 bytes, the actual size of the driver receive buffer is the value of settings.mReceiveBufferSize * 16.

8.2 The receiveBufferSize method

The receiveBufferSize method returns the size of the driver receive buffer, that is the value of the mReceiveBufferSize property of settings variable when the the begin method is called.

```
const uint32_t s = can.receiveBufferSize () ;
```

8.3 The receiveBufferCount method

The receiveBufferCount method returns the current number of messages in the driver receive buffer.

```
const uint32_t n = can.receiveBufferCount () ;
```

8.4 The receiveBufferPeakCount method


The receiveBufferPeakCount method returns the peak value of message count in the driver receive buffer.

```
const uint32_t max = can.receiveBufferPeakCount () ;
```

Note the driver receive buffer can overflow, if messages are not retrieved (by calling the receive or the dispatchReceivedMessage methods). If an overflow occurs, further calls of can.receiveBufferPeakCount () return can.receiveBufferSize ()+1.

9 Acceptance filters

It is recommended to read the Microchip documentation DS20001801H, section 4.5 page 33. The figure 5 shows the MCP2515 acceptance filter registers.



 $\mathbf{Figure} \ \mathbf{5} - \mathtt{MCP2515} \ \mathrm{acceptance} \ \mathrm{filters} \ (\mathtt{DS20001801H}, \ \mathrm{figure} \ 4.2 \ \mathrm{page} \ 25)$

9.1 Default behaviour

The can.begin (settings, [] can.isr ();) method sets the RXMO and RXM1 registers to 0, so, the MCP2515 receives all CAN bus messages.

More precisely, as RXMO is zero, all messages are received in RXBO. If a new message is received as RXBO is full, the new message is lost.

You can set the mRolloverEnable property of your ACAN2515Settings object to true (it is false by default). Doing that, if a new message is received as RXBO is full, the new message is transferred to RXB1. If RXB1 is full the new message is lost.

9.2 Defining filters

Sample sketch: the loopbackUsingFilters sketch shows how defining filters.

For defining filters, you should:

- define the values for the RXMO and RXM1 acceptance masks;
- submitting an ACAN2515AcceptanceFilter array to the ACAN2515::begin method.

The ACAN2515AcceptanceFilter array defines the values that the ACAN2515::begin method sets to the RXFi acceptance filter registers.

Four functions are available for managing filters:

- standard2515Mask and extended2515Mask functions for defining RXMi value;
- standard2515Filter and extended2515Filter functions for defining RXFi value.

RXMi and RXFi values you handle are ACAN2515Mask class instances, that provides four uint8_t properties: mSIDH, mSIDL, mEID8, mEID0. They correspond to the MCP2515 registers. If you want, you can set directly these properties, without using the above functions.

Filter remote and data frames. The MCP2515 filters do not handle the RTR bit: for example, you cannot specify you want to accept data frames and discard remote frames. This should be done by your code.

Multiple filter matches. From DS20001801H, section 4.5.4 page 34: If more than one acceptance filter matches, the FILHITn bits will encode the binary value of the lowest numbered filter that matched. For example, if filters, RXF2 and RXF4, match, the FILHITn bits will be loaded with the value for RXF2. This essentially prioritizes the acceptance filters with a lower numbered filter having higher priority. Messages are compared to filters in ascending order of filter number. This also ensures that the message will only be received into one buffer. This implies that RXB0 has a higher priority than RXB1.

The MCP2515 filters cannot be disabled, so all mask registers can be taken into account during the acceptance of a message. For example, if MCP2515 filters are defined with the RXMO, RXFO, RXF1 registers, leaving RXM1 equal to 0 provides the transfer to RXB1 of all messages discarded by RXFO and RXF1.

For dealing with all situations, the ACAN2515::begin method accepts three prototypes.

No filter.

```
ACAN2515Settings settings (QUARTZ_FREQUENCY, 125 * 1000);

const uint32_t errorCode = can.begin (settings, [] { can.isr (); });
```

No filter is provided, RXMO and RXM1 are set to 0, enabling the acceptance of all messages by RXBO.

One filter. For example:

Here, one type of message is accepted, extended (data or remote) frames with an identifier equal to 0x12345678. This defines explicitly RXMO and RXFO; for disabling acceptance by RXF1, it is set with RXFO value; RXM1 is set with RXMO value, and the RXF2 to RXF5 registers are set with the RXFO value. No message will be accepted by RXB1 filters.

The definition of a filter is associated with a call back function – here receive0. This function is called indirectly when the dispatchReceivedMessage method is called – see section 10 page 21.

Two filters. For example:

Here, two types of message are accepted, extended (data or remote) frames with an identifier equal to 0x12345678 or 0x18765432. This defines explicitly RXMO, RXFO and RXF1; RXM1 is set with RXMO value, and the RXF2 to RXF5 registers are set with the RXF1 value. No message will be accepted by RXB1 filters.

Three to five filters. For example, with four filters:

```
ACAN2515Settings settings (QUARTZ_FREQUENCY, 125 * 1000);
const ACAN2515Mask rxm0 = extended2515Mask (0x1FFFFFFF);
const ACAN2515Mask rxm1 = standard2515Mask (0x7FF, 0, 0);
const ACAN2515AcceptanceFilter filters [] = {
    {extended2515Filter (0x12345678), receive0}, // RXF0
    {extended2515Filter (0x18765432), receive1}, // RXF1
    {standard2515Filter (0x567, 0, 0), receive2}, // RXF2
    {standard2515Filter (0x123, 0, 0), receive3} // RXF3
};
```

Four types of message are accepted, extended (data or remote) frames with an identifier equal to 0x12345678 or 0x18765432, and standard (data or remote) frames with an identifier equal to 0x567 or 0x123. The RXF4 and RXF5 registers are set with the RXF3 value.

Six filters.

```
ACAN2515Settings settings (QUARTZ_FREQUENCY, 125 * 1000);
const ACAN2515Mask rxm0 = extended2515Mask (0x1FFFFFFF) ;
const ACAN2515Mask rxm1 = standard2515Mask (0x7FF, 0, 0);
const ACAN2515AcceptanceFilter filters [] = {
 {extended2515Filter (0x12345678), receive0}, // RXF0
 {extended2515Filter (0x18765432), receive1}, // RXF1
 {standard2515Filter (0x567, 0, 0), receive2},// RXF2
 {standard2515Filter (0x123, 0, 0), receive3},// RXF3
 {standard2515Filter (0x777, 0, 0), receive4},// RXF4
 {standard2515Filter (0x3AB, 0, 0), receive5} // RXF5
const uint32_t errorCode = can.begin (settings,
                                      [] { can.isr (); },
                                      rxm0, // Value set to RXMO register
                                      rxm1, // Value set to RXM1 register
                                      filters, // The filter array
                                      6); // Filter array size
```

Six types of message are accepted, all filter registers are explicitly defined.

9.2.1 Extended frames acceptance

The extended2515Mask and extended2515Filter functions helps you to define extended frame filters. Extended frame filters test extended identifier value.

The acceptance criterion is 7 :

```
acceptance_mask \& (received_identifier nXOR acceptance_filter) == 0
```

where & is the bit-wise and operator, and nXOR is the not xor bit-wise operator.

Accepting all extended frames.

```
const ACAN2515Mask rxm0 = extended2515Mask (0) ;
```

No extended frame identifier bit is tested, all extended frames are accepted.

⁷See DS20001801H, section 4.5 Message Acceptance Filters and Masks, page 33.

Accepting individual extended frames.

```
const ACAN2515Mask rxm0 = extended2515Mask (0x1FFFFFFF) ;
```

All extended frame identifier bits are tested, only extended frames whose identifiers match the filters are accepted.

Accepting several identifiers. The bits at 0 of the mask correspond to bits that are not tested for acceptance. For example:

```
const ACAN2515Mask rxm0 = extended2515Mask (0x1FFFFF0F) ;
```

If you define an acceptance filter by extended2515Filter (0x12345608), any extended frame with an identifier equal to 0x123456x8 is accepted.

9.2.2 Standard frames acceptance

The standard2515Mask and standard2515Filter functions helps you to define extended frame filters. Standard frame filters test standard identifier value, first and second data byte.

The acceptance criterion is⁸:

```
acceptance_mask \& ((received_identifier, data_byte_0, data_byte_1) nXOR acceptance_filter) == 0
```

where & is the bit-wise and operator, and nXOR is the not xor bit-wise operator.

Accepting all standard frames, without testing data bytes.

```
const ACAN2515Mask rxm0 = standard2515Mask (0, 0, 0);
```

Accepting individual standard frames, without testing data bytes.

```
const ACAN2515Mask rxm0 = standard2515Mask (0x7FF, 0, 0);
```

All standard frame identifier bits are tested, only standard frames whose identifiers match the filters are accepted.

Accepting several identifiers, without testing data bytes. The bits at 0 of the mask correspond to bits that are not tested for acceptance. For example:

```
const ACAN2515Mask rxm0 = standard2515Mask (0x70F, 0, 0);
```

If you define an acceptance filter by standard2515Filter (0x40A, 0, 0), any standard frame with an identifier equal to 0x4xA is accepted.

Filtering from first data byte. The second argument of standard2515Mask specify first data byte filtering. For example:

```
const ACAN2515Mask rxm0 = standard2515Mask (0x70F, 0xFF, 0) ;
```

If you define an acceptance filter by standard2515Filter (0x40A, 0x54, 0), any standard frame with an identifier equal to 0x4xA and first byte equal to 0x54 is accepted.

Empty standard frame. An empty standard frame (without any data byte) is accepted, the filtering condition on the first data byte is ignored (see loopbackFilterDataByte sample sketch).

⁸See DS20001801H, section 4.5 Message Acceptance Filters and Masks, page 33.

10 The dispatchReceivedMessage method

Sample sketch: the loopbackUsingFilters shows how using the dispatchReceivedMessage method.

Instead of calling the receive method, call the dispatchReceivedMessage method in your loop function. It calls the call back function associated with the matching filter.

If you have not defined any filter, do not use this function, call the receive method.

```
void loop () {
  can.dispatchReceivedMessage () ; // Do not use can.receive any more
  ...
}
```

The dispatchReceivedMessage method handles one message at a time. More precisely:

- if it returns false, the driver receive buffer was empty;
- if it returns true, the driver receive buffer was not empty, one message has been removed and dispatched.

So, the return value can used for emptying and dispatching all received messages:

```
void loop () {
  while (can.dispatchReceivedMessage ()) {
  }
  ...
}
```

If a filter definition does not name a call back function, the corresponding messages are lost.

The dispatchReceivedMessage method has an optional argument – NULL by default: a function name. This function is called for every message that pass the receive filters, with an argument equal to the matching filter index:

```
void filterMatchFunction (const uint32_t inFilterIndex) {
    ...
}

void loop () {
    can.dispatchReceivedMessage (filterMatchFunction) ;
    ...
}
```

You can use this function for maintaining statistics about receiver filter matches.

11 The ACAN2515::begin method reference

11.1 The ACAN2515::begin method prototypes

There are three begin method prototypes:

11.2 Defining explicitly the interrupt service routine

In this document, the *interrupt service routine* is defined by a lambda expression:

```
const uint32_t errorCode = can.begin (settings, [] { can.isr () ; }) ;
```

Instead of a lambda expression, you are free to define the interrupt service routine as a function:

```
void canISR () {
   can.isr ();
}
```

And you pass canISR as argument to the begin method:

```
const uint32_t errorCode = can.begin (settings, canISR) ;
```

11.3 The error code

The ACAN2515::begin method returns an error code. The value 0 denotes no error. Otherwise, you consider every bit as an error flag, as described in table 1. An error code could report several errors. The ACAN2515 class defines static constants for naming errors.

| \mathbf{Bit} | Static constant Name | Link |
|----------------|---|----------------------------|
| 0 | kNoMCP2515 | section $11.3.1$ page 23 |
| 1 | ${\tt kTooFarFromDesiredBitRate}$ | section $11.3.2$ page 23 |
| 2 | ${\tt kInconsistentBitRateSettings}$ | section $11.3.3$ page 23 |
| 3 | kINTPinIsNotAnInterrupt | section $11.3.4$ page 23 |
| 4 | kISRIsNull | section $11.3.5$ page 24 |
| 5 | ${\tt kRequestedModeTimeOut}$ | section $11.3.6$ page 24 |
| 6 | kAcceptanceFilterArrayIsNULL | section $11.3.7$ page 24 |
| 7 | ${\tt kOneFilterMaskRequiresOneOrTwoAcceptanceFilters}$ | section $11.3.8$ page 24 |
| 8 | ${\tt kTwoFilterMasksRequireThreeToSixAcceptanceFilters}$ | section $11.3.9$ page 24 |

Table 1 - The ACAN2515::begin method error code bits

11.3.1 kNoMCP2515

The ACAN2515::begin method checks accessibility by writing and reading back the CNF1_REGISTER first with the 0x55 value, then with the 0xAA value. This error is raised when the read value is different from the written one. It means that the MCP2515 cannot be accessed via SPI.

11.3.2 kTooFarFromDesiredBitRate

This error occurs when the mBitRateClosedToDesiredRate property of the settings object is false. This means that the ACAN2515Settings constructor cannot compute a CAN bit configuration close enough to the desired bit rate. For example:

```
void setup () {
   ACAN2515Settings settings (QUARTZ_FREQUENCY, 1); // 1 bit/s !!!
   // Here, settings.mBitRateClosedToDesiredRate is false
   const uint32_t errorCode = can.begin (settings, [] { can.isr (); });
   // Here, errorCode contains ACAN2515::kCANBitConfigurationTooFarFromDesiredBitRate
}
```

11.3.3 kInconsistentBitRateSettings

The ACAN2515Settings constructor allways returns consistent bit rate settings – even if the settings provide a bit rate too far away the desired bit rate. So this error occurs only when you have changed the CAN bit properties (mBitRatePrescaler, mPropagationSegment, mPhaseSegment1, mPhaseSegment2, mSJW), and one or more resulting values are inconsistent. See section 12.2 page 28.

11.3.4 kINTPinIsNotAnInterrupt

The pin you provide for handling the MCP2515 interrupt has no interrupt capability.

11.3.5 kISRIsNull

The interrupt service routine argument is NULL, you should provide a valid function.

11.3.6 kRequestedModeTimeOut

During configuration by the ACAN2515::begin method, the MCP2515 is in the *configuration* mode. At this end of this process, the mode specified by the inSettings.mRequestedMode value is requested. The switch to this mode is not immediate, a register is repetitively read for checking the switch is done. This error is raised if the switch is not completed within a delay between 1 ms and 2 ms.

11.3.7 kAcceptanceFilterArrayIsNULL

The ACAN2515::begin method you have called names the inAcceptanceFilters argument, but it is NULL.

$11.3.8 \quad \verb"kOneFilterMaskRequiresOneOrTwoAcceptanceFilters"$

The ACAN2515::begin method you have called names the inRXMO argument (but not inRXM1), you should provide the value 1 or 2 to the inAcceptanceFilterCount argument.

11.3.9 kTwoFilterMasksRequireThreeToSixAcceptanceFilters

The ACAN2515::begin method you have called names the inRXMO and the the inRXM1 arguments, you should provide the value 3 to 6 to the inAcceptanceFilterCount argument.

12 ACAN2515Settings class reference

Note. The ACAN2515Settings class is not Arduino specific. You can compile it on your desktop computer with your favorite C++ compiler. In the https://github.com/pierremolinaro/acan2515-dev GitHub repository, a command line tool is defined for exploring all CAN bit rates from 1 bit/s and 20 Mbit/s for a 16 MHz quartz: 63810 bit rates are valid, and 29 are exact. It also checks that computed CAN bit decompositions are all consistent, even if they are too far from the desired baud rate.

12.1 The ACAN2515Settings constructor: computation of the CAN bit settings

The constructor of the ACAN2515Settings has two mandatory arguments: the quartz frequency, and the desired bit rate. It tries to compute the CAN bit settings for this bit rate. If it succeeds, the constructed object has its mBitRateClosedToDesiredRate property set to true, otherwise it is set to false. For example:

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
```

```
ACAN2515Settings settings (QUARTZ_FREQUENCY, 1 * 1000 * 1000); // 1 Mbit/s
// Here, settings.mBitRateClosedToDesiredRate is true
...
}
```

Of course, with a 16 MHz quartz, CAN bit computation allways succeeds for classical bit rates: 1 Mbit/s, 500 kbit/s, 250 kbit/s, 125 kbit/s. But CAN bit computation can also succeed for some unusual bit rates, as 727 kbit/s. You can check the result by computing actual bit rate, and the distance from the desired bit rate:

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 727 * 1000) ; // 727 kbit/s
    Serial.print ("mBitRateClosedToDesiredRate:__") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 1 (--> is true)
    Serial.print ("actual_ubit_rate:__") ;
    Serial.println (settings.actualBitRate ()) ; // 727272 bit/s
    Serial.print ("distance:__") ;
    Serial.println (settings.ppmFromDesiredBitRate ()) ; // 375 ppm
    ...
}
```

The actual bit rate is 727,272 bit/s, and its distance from desired bit rate is 375 ppm. "ppm" stands for "part-per-million", and 1 ppm = 10^{-6} . In other words, 10,000 ppm = 1%.

By default, a desired bit rate is accepted if the distance from the computed actual bit rate is lower or equal to 1,000~ppm = 0.1~%. You can change this default value by adding your own value as third argument of ACAN2515Settings constructor:

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 727 * 1000, 100) ;
    Serial.print ("mBitRateClosedToDesiredRate:__") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 0 (--> is false)
    Serial.print ("actual_bit_rate:_") ;
    Serial.println (settings.actualBitRate ()) ; // 727272 bit/s
    Serial.print ("distance:_") ;
    Serial.println (settings.ppmFromDesiredBitRate ()) ; // 375 ppm
    ...
}
```

The third argument does not change the CAN bit computation, it only changes the acceptance test for setting the mBitRateClosedToDesiredRate property. For example, you can specify that you want the computed actual bit to be exactly the desired bit rate:

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
   ...
```

```
ACAN2515Settings settings (QUARTZ_FREQUENCY, 500 * 1000, 0); // Max distance is 0 ppm Serial.print ("mBitRateClosedToDesiredRate:"); Serial.println (settings.mBitRateClosedToDesiredRate); // 1 (--> is true) Serial.print ("actual_ubit_rate:"); Serial.println (settings.actualBitRate ()); // 500,000 bit/s Serial.print ("distance:"); Serial.println (settings.ppmFromDesiredBitRate ()); // 0 ppm ...
}
```

In any way, the bit rate computation allways gives a consistent result, resulting an actual bit rate closest from the desired bit rate. For example:

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 440 * 1000) ; // 440 kbit/s
    Serial.print ("mBitRateClosedToDesiredRate:") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 0 (--> is false)
    Serial.print ("actual_ubit_urate:_") ;
    Serial.println (settings.actualBitRate ()) ; // 444,444 bit/s
    Serial.print ("distance:_") ;
    Serial.println (settings.ppmFromDesiredBitRate ()) ; // 10,100 ppm
    ...
}
```

You can get the details of the CAN bit decomposition. For example:

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
 ACAN2515Settings settings (QUARTZ_FREQUENCY, 440 * 1000); // 440 kbit/s
 Serial.print ("mBitRateClosedToDesiredRate:");
 Serial.println (settings.mBitRateClosedToDesiredRate) ; // 0 (--> is false)
 Serial.print ("actual_bit_rate:_");
 Serial.println (settings.actualBitRate ()); // 444,444 bit/s
 Serial.print ("distance:");
 Serial.println (settings.ppmFromDesiredBitRate ()); // 10,100 ppm
 Serial.print ("Biturateuprescaler:u");
 Serial.println (settings.mBitRatePrescaler) ; // BRP = 1
 Serial.print ("Propagation segment: ");
 Serial.println (settings.mPropagationSegment) ; // PropSeg = 6
 Serial.print ("Phase segment 1: ");
 Serial.println (settings.mPhaseSegment1) ; // PS1 = 5
 Serial.print ("Phase segment 2: ");
 Serial.println (settings.mPhaseSegment2); // PS2 = 6
 Serial.print ("Resynchronization_Jump_Width:__");
 Serial.println (settings.mSJW); // SJW = 4
```

```
Serial.print ("Triple_Sampling:_");
Serial.println (settings.mTripleSampling); // 0, meaning single sampling
Serial.print ("Sample_Point:_");
Serial.println (settings.samplePointFromBitStart ()); // 68, meaning 68%
Serial.print ("Consistency:_");
Serial.println (settings.CANBitSettingConsistency ()); // 0, meaning 0k
...
}
```

The samplePointFromBitStart method returns sample point, expressed in per-cent of the bit duration from the beginning of the bit.

Note the computation may calculate a bit decomposition too far from the desired bit rate, but it is allways consistent. You can check this by calling the CANBitSettingConsistency method.

You can change the property values for adapting to the particularities of your CAN network propagation time. By example, you can increment the mPhaseSegment1 value, and decrement the mPhaseSegment2 value in order to sample the CAN Rx pin later.

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 500 * 1000) ; // 500 kbit/s
    Serial.print ("mBitRateClosedToDesiredRate:__") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 1 (--> is true)
    settings.mPhaseSegment1 ++ ; // 5 -> 6: safe, 1 <= PS1 <= 8
    settings.mPhaseSegment2 -- ; // 5 -> 4: safe, 2 <= PS2 <= 8 and SJW <= PS2
    Serial.print ("Sample_Point:__") ;
    Serial.println (settings.samplePointFromBitStart ()) ; // 75, meaning 75%
    Serial.println (settings.actualBitRate ()) ; // 500000: ok, bit rate did not change
    Serial.println (settings.actualBitRate ()) ; // 500000: ok, bit rate did not change
    Serial.println (settings.CANBitSettingConsistency ()) ; // 0, meaning 0k
    ...
}</pre>
```

Be aware to allways respect CAN bit timing consistency! The constraints are:

```
1\leqslant \texttt{mBitRatePrescaler}\leqslant 64 1\leqslant \texttt{mSJW}\leqslant 4 1\leqslant \texttt{mPropagationSegment}\leqslant 8 Single sampling: 1\leqslant \texttt{mPhaseSegment1}\leqslant 8 Triple sampling: 2\leqslant \texttt{mPhaseSegment1}\leqslant 8 2\leqslant \texttt{mPhaseSegment2}\leqslant 8 \texttt{mSJW}< \texttt{mPhaseSegment2} \texttt{mPhaseSegment2}
```

Resulting actual bit rate is given by:

```
\label{eq:Actual} \mbox{Actual bit rate} = \frac{QuartzFrequency \; / \; 2}{\mbox{\tt mBitRatePrescaler} \cdot (1 + \mbox{\tt mPropagationSegment} + \mbox{\tt mPhaseSegment1} + \mbox{\tt mPhaseSegment2})}
```

And sampling points (in per-cent unit) are given by:

```
Sampling \ point \ (\textit{single sampling}) = 100 \cdot \frac{1 + \texttt{mPropagationSegment} + \texttt{mPhaseSegment1}}{1 + \texttt{mPropagationSegment} + \texttt{mPhaseSegment1} + \texttt{mPhaseSegment2}} Sampling \ first \ point \ (\textit{triple sampling}) = 100 \cdot \frac{\texttt{mPropagationSegment} + \texttt{mPhaseSegment1}}{1 + \texttt{mPropagationSegment} + \texttt{mPhaseSegment1}}
```

12.2 The CANBitSettingConsistency method

This method checks the CAN bit decomposition (given by mBitRatePrescaler, mPropagationSegment, mPhaseSegment1, mPhaseSegment2, mSJW property values) is consistent.

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 500 * 1000) ; // 500 kbit/s
    Serial.print ("mBitRateClosedToDesiredRate:__") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 1 (--> is true)
    settings.mPhaseSegment1 = 0 ; // Error, mPhaseSegment1 should be >= 1 (and <= 8)
    Serial.print ("Consistency:__0x") ;
    Serial.println (settings.CANBitSettingConsistency (), HEX) ; // 0x10, meaning error
    ...
}</pre>
```

The CANBitSettingConsistency method returns 0 if CAN bit decomposition is consistent. Otherwise, the returned value is a bit field that can report several errors – see table 2.

The ACAN2515Settings class defines static constant properties that can be used as mask error. For example:

```
public: static const uint32_t kBitRatePrescalerIsZero = 1 << 0 ;</pre>
```

| \mathbf{Bit} | Error Name | Error |
|----------------|---|---|
| 0 | kBitRatePrescalerIsZero | ${\tt mBitRatePrescaler} == 0$ |
| 1 | ${\tt kBitRatePrescalerIsGreaterThan 64}$ | ${\tt mBitRatePrescaler} > 64$ |
| 2 | ${\tt kPropagationSegmentIsZero}$ | ${\tt mPropagationSegment} == 0$ |
| 3 | ${\tt kPropagationSegmentIsGreaterThan8}$ | ${\tt mPropagationSegment} > 8$ |
| 4 | kPhaseSegment1IsZero | ${\tt mPhaseSegment1} == 0$ |
| 5 | kPhaseSegment1IsGreaterThan8 | ${\tt mPhaseSegment1} > 8$ |
| 6 | kPhaseSegment2IsLowerThan2 | ${\tt mPhaseSegment2} < 2$ |
| 7 | kPhaseSegment2IsGreaterThan8 | ${\tt mPhaseSegment2} > 8$ |
| 8 | ${\tt kPhaseSegment1Is1AndTripleSampling}$ | $({\tt mPhaseSegment1} == 1) \ \&\& \ {\tt mTripleSampling}$ |
| 9 | kSJWIsZero | $\mathtt{mSJW} == 0$ |
| 10 | kSJWIsGreaterThan4 | $\mathtt{mSJW} > 4$ |
| 11 | ${\tt kSJWIsGreaterThanOrEqualToPhaseSegment2}$ | $\mathtt{mSJW} >= \mathtt{mPhaseSegment2}$ |
| 12 | ${\tt kPhaseSegment2IsGreaterThanPSPlusPS1}$ | ${\tt mPhaseSegment2} > ({\tt mPropagationSegment} + {\tt mPhaseSegment1})$ |

Table 2 - The ACAN2515Settings::CANBitSettingConsistency method error codes

12.3 The actualBitRate method

The actualBitRate method returns the actual bit computed from mBitRatePrescaler, mPropagationSegment, mPhaseSegment1, mPhaseSegment2, mSJW property values.

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 440 * 1000) ; // 440 kbit/s
    Serial.print ("mBitRateClosedToDesiredRate:__") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 0 (--> is false)
    Serial.print ("actual_bit_rate:_") ;
    Serial.println (settings.actualBitRate ()) ; // 444,444 bit/s
    ...
}
```

Note. If CAN bit settings are not consistent (see section 12.2 page 28), the returned value is irrelevant.

12.4 The exactBitRate method

The exactBitRate method returns true if the actual bit rate is equal to the desired bit rate, and false otherwise.

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 727 * 1000) ; // 727 kbit/s
    Serial.print ("mBitRateClosedToDesiredRate:") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 1 (--> is true)
    Serial.print ("actual_bit_rate:") ;
    Serial.println (settings.actualBitRate ()) ; // 727272 bit/s
    Serial.print ("distance:") ;
    Serial.println (settings.ppmFromDesiredBitRate ()) ; // 375 ppm
    Serial.print ("Exact:") ;
    Serial.println (settings.exactBitRate ()) ; // 0 (---> false)
    ...
}
```

Note. If CAN bit settings are not consistent (see section 12.2 page 28), the returned value is irrelevant.

For a 16 MHz clock, the 28 exact bit rates are: 5 kbit/s, 6250 bit/s, 6400 bit/s, 8 kbit/s, 10 kbit/s, 12500 bit/s, 12800 bit/s, 15625 bit/s, 16 kbit/s, 20 kbit/s, 25 kbit/s, 31250 bit/s, 32 kbit/s, 40 kbit/s, 50 kbit/s, 62500 bit/s, 64 kbit/s, 80 kbit/s, 100 kbit/s, 125 kbit/s, 160 kbit/s, 200 kbit/s, 250 kbit/s, 320 kbit/s, 400 kbit/s, 500 kbit/s, 800 kbit/s, 1000 kbit/s.

For a 10 MHz clock, the 24 exact bit rates are: 3125 bit/s, 4 kbit/s, 5 kbit/s, 6250 bit/s, 8 kbit/s, 10 kbit/s, 12500 bit/s, 15625 bit/s, 20 kbit/s, 25 kbit/s, 31250 bit/s, 40 kbit/s, 50 kbit/s, 62500 bit/s, 78125 bit/s, 100 kbit/s, 125 kbit/s, 156250 bit/s, 200 kbit/s, 250 kbit/s, 312500 bit/s, 500 kbit/s, 625 kbit/s, 1000 kbit/s.

For a 8 MHz clock, the 28 exact bit rates are: 2500 bit/s, 3125 bit/s, 3200 bit/s, 4 kbit/s, 5 kbit/s, 6250 bit/s, 6400 bit/s, 8 kbit/s, 10 kbit/s, 12500 bit/s, 15625 bit/s, 16 kbit/s, 20 kbit/s, 25 kbit/s, 31250 bit/s, 32 kbit/s, 40 kbit/s, 50 kbit/s, 62500 bit/s, 80 kbit/s, 100 kbit/s, 125 kbit/s, 160 kbit/s, 200 kbit/s, 250 kbit/s, 400 kbit/s, 500 kbit/s, 800 kbit/s.

Note an 1 Mbit/s bit rate cannot be performed with a 8 MHz clock.

12.5 The ppmFromDesiredBitRate method

The ppmFromDesiredBitRate method returns the distance from the actual bit rate to the desired bit rate, expressed in part-per-million (ppm): 1 ppm = 10^{-6} . In other words, 10,000 ppm = 1%.

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 727 * 1000) ; // 727 kbit/s
    Serial.print ("mBitRateClosedToDesiredRate:__") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 1 (--> is true)
    Serial.print ("actual__bit__rate:__") ;
    Serial.println (settings.actualBitRate ()) ; // 727272 bit/s
    Serial.print ("distance:__") ;
```

```
Serial.println (settings.ppmFromDesiredBitRate ()); // 375 ppm ...
}
```

Note. If CAN bit settings are not consistent (see section 12.2 page 28), the returned value is irrelevant.

12.6 The samplePointFromBitStart method

The samplePointFromBitStart method returns the distance of sample point from the start of the CAN bit, expressed in part-per-cent (ppc): 1 ppc = $1\% = 10^{-2}$. If triple sampling is selected, the returned value is the distance of the first sample point from the start of the CAN bit. It is a good practice to get sample point from 65% to 80%.

```
const uint32_t QUARTZ_FREQUENCY = 16 * 1000 * 1000 ; // 16 MHz
void setup () {
    ...
    ACAN2515Settings settings (QUARTZ_FREQUENCY, 500 * 1000) ; // 500 kbit/s
    Serial.print ("mBitRateClosedToDesiredRate:") ;
    Serial.println (settings.mBitRateClosedToDesiredRate) ; // 1 (--> is true)
    Serial.print ("Sample_point:") ;
    Serial.println (settings.samplePointFromBitStart ()) ; // 68 --> 68%
    ...
}
```

Note. If CAN bit settings are not consistent (see section 12.2 page 28), the returned value is irrelevant.

12.7 Properties of the ACAN2515Settings class

All properties of the ACAN2515Settings class are declared public and are initialized (table 3). The default values of properties from mDesiredBitRate until mTripleSampling corresponds to a CAN bit rate of QUARTZ_FREQUENCY / 64, that is 250,000 bit/s for a 16 MHz quartz.

12.7.1 The mOneShotModeEnabled property

This boolean property corresponds to the OSM bit of the CANCTRL control register. It is false by default.

12.7.2 The mTXBPriority property

This property defines the transmit priority associated the $\mathtt{TXB}i$ registers:

- bits 1-0: priority of TXBO;
- bits 3-2: priority of TXB1;
- bits 5-4: priority of TXB2;

| Property | Type | Initial value | Comment |
|-------------------------------------|-----------------------|------------------------------------|----------------------------|
| t mQuartzFrequency | uint32_t | QUARTZ_FREQUENCY | |
| ${\tt mDesiredBitRate}$ | uint32_t | ${\tt QUARTZ_FREQUENCY} \ / \ 64$ | |
| mBitRatePrescaler | uint8_t | 2 | See section 12.1 page 24 |
| ${\tt mPropagationSegment}$ | uint8_t | 5 | See section 12.1 page 24 |
| mPhaseSegment1 | uint8_t | 5 | See section 12.1 page 24 |
| mPhaseSegment2 | uint8_t | 5 | See section 12.1 page 24 |
| mSJW | uint8_t | 4 | See section 12.1 page 24 |
| mTripleSampling | bool | false | See section 12.1 page 24 |
| ${\tt mBitRateClosedToDesiredRate}$ | bool | true | See section 12.1 page 24 |
| ${\tt mOneShotModeEnabled}$ | bool | false | See section 12.7.1 page 31 |
| mTXBPriority | uint8_t | 0 | See section 12.7.2 page 31 |
| ${\tt mRequestedMode}$ | ACAN2515RequestedMode | NormalMode | See section 12.7.3 page 32 |
| mCLKOUT_SOF_pin | ACAN2515CLKOUT_SOF | CLOCK | See section 12.7.4 page 32 |
| mRolloverEnable | bool | true | See section 12.7.5 page 32 |
| ${\tt mReceiveBufferSize}$ | uint16_t | 32 | See section 8.1 page 15 |
| ${\tt mTransmitBufferOSize}$ | uint16_t | 16 | See section 7.2 page 13 |
| ${\tt mTransmitBuffer1Size}$ | uint16_t | 0 | See section 7.2 page 13 |
| mTransmitBuffer2Size | uint16_t | 0 | See section 7.2 page 13 |

Table 3 - Properties of the ACAN2515Settings class

• bits 7-6: unused.

By default, its value is 0, all three TXBi registers get the same 0 priority.

12.7.3 The mRequestedMode property

This property defines the mode requested at this end of the configuration: NormalMode (default value), ListenOnlyMode, LoopBackMode.

12.7.4 The mCLKOUT property

This property defines signal output on the CLKOUT/SOF pin; possible values are: CLOCK (default value), CLOCK2, CLOCK4, CLOCK8, SOF, Hiz.

12.7.5 The mRolloverEnable property

This boolean property corresponds to the BUKT bit of the RXBOCTRL control register. If true (value by default), RXBO message will roll over and be written to RXB1 if RXBO is full; if false, rollover is disabled.

13 CAN controller state

Two methods return the receive error counter and the transmit error counter.

13.1 The receiveErrorCounter method

```
public: uint8_t receiveErrorCounter (void) ;
```

13.2 The transmitErrorCounter method

```
public: uint8_t transmitErrorCounter (void) ;
```